



INSTITUT UNIVERSITAIRE DE TECHNOLOGIE

UNIVERSITE DE BRETAGNE-SUD

Licence Professionnelle Conception des Systèmes Décisionnels

**PROJET SAS :
CREATION D'INTERFACES WEB ET DE RAPPORTS
AVEC LE LOGICIEL SAS 9**



Document réalisé par Julien CHANU

Année universitaire 2008-2009



INSTITUT UNIVERSITAIRE DE TECHNOLOGIE

UNIVERSITE DE BRETAGNE-SUD

Licence Professionnelle Conception des Systèmes Décisionnels

**PROJET SAS :
CREATION D'INTERFACES WEB ET DE RAPPORTS
AVEC LE LOGICIEL SAS 9**



Document réalisé par Christelle GENTRIC et Julien CHANU

Année universitaire 2008-2009

*Remerciements à M. Erwan PRUD'HOMME qui
a bien voulu encadrer notre projet, pour son aide
et ses renseignements précieux qu'il nous a
fourni.*

Vannes, décembre 2008

SOMMAIRE

1. INTRODUCTION.....	5
1.1 – Présentation du projet SAS	5
1.2 – Objectifs du projet SAS.....	5
2. LES TECHNIQUES UTILISÉES POUR LE REPORTING.....	7
2.1 – La procédure TABULATE.....	7
2.2 – La procédure REPORT	7
2.3 – L’ODSOUT avec l’étape DATA.....	8
3. MANUEL D’UTILISATEUR.....	11
3.1 – Installation de l’application web	11
3.2 – Interface Homme/Machine.....	12
4. CONCLUSION	16
5. ANNEXES	18
5.1 – Annexe 1 : Dictionnaire des données.....	18
5.2 – Annexe 2 : Code du programme SAS	20
5.3 – Annexe 3 : Architecture de l’application web.....	34
5.4 – Annexe 4 : Références.....	35

1. INTRODUCTION

1. INTRODUCTION

1.1 – Présentation du projet SAS

Le but du projet est d'aider un vendeur de véhicules de sport, qui souhaite rassurer sa clientèle sur un investissement pérenne. L'idée de base est que le commercial assure la sécurité de ses clients, dans l'acquisition d'une voiture de sport ou de haut de gamme, qui peut-être un placement plus ou moins à long terme pour un client ; sachant que ce financement peut s'avouer être imprudent ; en s'appuyant sur une base de données comprenant des véhicules de collection, de sport ou de haut de gamme. L'information de ces véhicules sera constituée d'une partie sur la description de l'automobile et d'autre part, la tendance de ses valeurs financières au cours du temps, dont son coût à la commercialisation et de ses différentes aux enchères ; pour cette même voiture ou semblable à celle qui est présentée.

1.2 – Objectifs du projet SAS

Le projet consiste à réaliser une application dynamique au travers de plusieurs pages web, listant les différents véhicules de collection, enregistrés dans une base de données. En présentant, leurs différentes caractéristiques primaires. Pour définir chacune de ces voitures, il convient en outre de les présenter, et d'y rattacher des éléments clés concernant leur valeur. Pour cette étude, il a été convenu d'y ajouter une mise en forme, la plus adaptée possible à cette application.

Plusieurs contraintes ont été affectées à la réalisation de ce projet. Dans un premier temps, les éléments de reporting doivent uniquement utiliser trois techniques distinctes, qui sont : les procédures TABULATE et REPORT, ainsi que l'ODSOUT dans une étape DATA ; du module SAS/BASE. Ces techniques seront définies dans le prochain chapitre. D'autre part, l'utilisation de la procédure PRINT, du module SAS/BASE, n'est pas autorisée pour la conception de cet ouvrage.

Pour ce projet, nous ne conserverons que les automobiles, dont nous connaissons leurs caractéristiques et leurs valeurs aux enchères. Chaque voiture sera représentée par une table, qui contiendra sa description et ses différentes valeurs financières de début (coûts initiaux) et de fin (coûts des enchères). Une page web sera composée de plusieurs lignes, sachant qu'une ligne correspondra à une voiture, en fonction de son année de mise en circulation avec le coût à la commercialisation de celle-ci, et la moyenne des coûts des véhicules vendus aux enchères. Les valeurs financières seront exprimées en euros constant (€). De plus, on y déterminera la progression de sa valeur (en %), entre son coût initial et la moyenne des coûts des voitures semblables à celle-ci vendus aux enchères ; ainsi que son délai de conservation entre l'année de mise en circulation et sa date aux enchères. Toutes ces informations seront affichées au sein d'un tableau de données.

Pour la réalisation de cette application, il a fallu créer une macro qui devait concevoir autant de rapports que de véhicules complets ; c'est-à-dire ses données initiales et enchères, ainsi que ses éléments clés primaires. Ces rapports figureront sur des feuilles html distinctes et comporter un nom unique (xxx.html), pour chaque automobile. Cette application sera constituée de plusieurs liens distincts entre le document initial et les documents finaux.

2. LES TECHNIQUES UTILISÉES POUR LE REPORTING

2. LES TECHNIQUES UTILISÉES POUR LE REPORTING

Pour la réalisation du projet, nous étions contraints à utiliser seulement que trois techniques de reporting, qui sont les procédures TABULATE et REPORT, ainsi que l'ODSOUT à l'aide d'une étape DATA ; du module de SAS/BASE.

2.1 – La procédure TABULATE

La procédure TABULATE du module SAS/BASE permet l'édition de statistiques descriptives sous forme d'un tableau croisé et avec une mise en page soignée. Les tableaux peuvent être dimensionnés selon trois axes, qui sont PAGE, LIGNE et COLONNE. Chaque axe peut combiner différentes variables selon différents modes. Le tableau se construit en calculant certaines statistiques pour une ou plusieurs variables d'analyse (de type continu), pour chaque sous-ensemble déterminé par la combinaison des variables de classification (discrètes avec peu de modalités). Chaque cellule correspond finalement à une statistique calculée pour une variable d'analyse, pour un certain sous-ensemble. Chaque statistique peut spécifier un format d'impression et une étiquette, qui peuvent être le nombre d'observation ayant une valeur non manquante pour la variable d'analyse (N), la valeur minimale (MIN), la valeur maximale (MAX), la somme des valeurs de la variable d'analyse (SUM), la moyenne arithmétique (MEAN), la variance (VAR), l'écart-type (STD), le pourcentage des fréquences (PCTN), Les variables de classification peuvent être reliées entre elles par trois opérations, qui sont l'imbrication, la concaténation liste simple et le groupage.

La procédure TABULATE regroupe les fonctionnalités des procédures MEANS et FREQ du module SAS/BASE. Si sa syntaxe est plus complexe, elle permet également la réalisation d'un tableau statistique « sur mesure » en fonction de besoins. Il s'agit donc de la procédure incontournable pour la réalisation d'un tableau de bord, par exemple, ou de tableaux destinés à un rapport. Pour conclure, c'est la procédure, la plus simple pour la réalisation de tableaux réellement croisés, quand il y a une ou plusieurs variables de classification qui apparaissent en colonnes. Celle-ci permet également de présenter les statistiques sur plusieurs lignes, ce qui n'est pas le cas de la procédure REPORT.

2.2 – La procédure REPORT

La procédure REPORT, du module SAS/BASE, combine les caractéristiques des procédures PRINT, MEANS et TABULATE, avec des fonctions de l'étape DATA. La procédure REPORT peut-être utilisé de trois façons différentes, qui sont :

- dans un environnement Windows, avec une installation qui invite les utilisateurs à construire leur rapport.
- dans un environnement Windows, sans l'impulsion des installations.
- dans un environnement n'utilisant pas Windows. Dans ce cas, l'utilisateur est soumis à une série de déclarations avec la procédure REPORT, comme il le fait dans d'autres procédures de SAS/BASE.

La procédure REPORT est agréable, si la logique du tableau à produire est organisée par lignes. Celle-ci permet de personnaliser les synthèses, de calculer de nouvelles variables instantanément (en évitant des procédures et des étapes DATA intermédiaires, ce qui est appréciable sur des gros volumes de données), de pondérer séparément toutes les variables d'analyse. Enfin, les styles peuvent s'appliquer par cellule, par colonne ou par ligne, de manière conditionnelle (la procédure TABULATE ne propose de la mise en forme conditionnelle que par cellule). Pour finir, la procédure REPORT est un outil qui est certainement très puissant, mais peut vite devenir très compliqué. De notre expérience, il apparaît plus judicieux de restreindre l'usage de la procédure REPORT à la diffusion de résultats préalablement agencés, en triant les données avec des variables numériques non affichées et ne montrant que les variables textes.

2.3 – L'ODSOUT avec l'étape DATA

L'étape DATA propose depuis longtemps la construction de rapport sur mesure, à l'aide des instructions FILE, PRINT et PUT. Avec l'ODS (Output Delivery System), une instruction FILE, PRINT ODS= (STYLE=Template) était apparue, à utiliser conjointement avec l'instruction PUT_ODS ; pour produire avec une étape DATA des sorties tabulaires. Cependant, la flexibilité de l'étape DATA ne permettait pas de résoudre tous les problèmes de formatage complexe de sorties, en particulier les cellules fusionnées. Il a donc été ajouté, de manière expérimentale en version 9.1.3, aux fonctionnalités de l'étape DATA un objet ODSOUT, qui représente un rapport, et dont le remplissage, à l'aide de méthodes définissant des contenants emboîtés, est d'une grande souplesse. En revanche, cette étape DATA fonctionne les principales destinations ODS telles que HTML (HyperText Markup Language, page web), RTF (Rich Text Format, fichier texte) et PDF (Portable Document Format).

Les contenants que l'on décrit dans un objet ODSOUT sont tabulaires par nature. Il faut donc les penser comme tels, et repérer à l'avance où se situeront les fusions de cellules, verticalement et/ou horizontalement. Leur organisation est hiérarchique, le contenant le plus élémentaires étant emboîté dans un contenant parent, et ainsi de suite (cf. Figure 1).




Schéma 1 – Arborescence des contenants dans l'objet ODSOUT

L'objet ODSOUT peut contenir un ou plusieurs tableaux. Ceux-ci peuvent courir sur une ou plusieurs pages. Chacun de ces tableaux est composé de lignes, dont certaines peuvent être de type « en-tête ». Dans chaque ligne, on construit des cellules qui sont des contenants élémentaires. On peut définir des titres qui s'insèrent avant le tableau et des textes qui s'ajoutent avant ou après le tableau selon son emplacement.

L'ODSOUT est souvent utilisé dans des cas désespérés, déjà parce que la syntaxe est expérimentale (donc susceptible d'être totalement transformée lors d'un changement de version de SAS) et qu'en plus, elle est très redondante. Son énorme avantage, c'est la fusion des cellules sur plusieurs lignes ou colonnes. Elle permet de rendre proprement des pages de textes, des tableaux très complexes, le genre de choses qu'on ferait "à l'ancienne" avec des FILE et des PUT, mais le résultat est laid via l'ODS.

3. MANUEL D'UTILISATEUR

3. MANUEL D'UTILISATEUR

3.1 – Installation de l'application web

- Placer le fichier **Voiture de collection.zip** dans le répertoire choisi.
- Décompresser le fichier **Voiture de collection.zip**.
- Ouvrir le fichier **Voiture.sas** avec le logiciel SAS 9, ou versions antérieures.
- Modifier le *chemin* de la macro-variable « *chemin*, » en saisissant l'adresse *source* où se situent les fichiers et les dossiers décompressés du fichier **Voiture de collection.zip**, à la ligne 2 du programme. Par exemple, ici, nous utiliserons l'adresse suivante : *C:\Users\Julien\Desktop\Voiture de collection*.




Figure 1 – Interface graphique du dossier *Voiture de collection*

- Exécuter le programme **Voiture.sas** avec le logiciel SAS 9, ou versions antérieures.
- A la fin de l'exécution du programme **Voiture.sas**, l'application est installée.

3.2 – Interface Homme/Machine

Pour lancer l'application, il faut aller dans le dossier **Sortie**. Ensuite, il faut ouvrir le fichier **index.html**. Et vous obtenez ceci :



Figure 2 – Interface graphique du fichier *index.html*

Ensuite, pour accéder aux informations des différentes voitures de collection. Il faut cliquer sur *Entrez* ou sur *l'impact de la voiture*. Une nouvelle page s'affiche avec les marques de voitures disponibles.




Figure 3 – Liste des marques de voitures

Ensuite, l'utilisateur pourra accéder à la liste des différents modèles de voitures, après avoir choisi la marque de voiture qu'il l'intéresse, tout en cliquant sur le nom de la marque ou le logo de la marque. Une liste de modèles de voitures pour une marque concernée se présente :




Figure 4 – Liste des modèles d'une marque de voiture (ici Ferrari)

Pour s'informer sur la description et sur les valeurs financières d'une voiture, l'utilisateur clique sur le modèle de la marque de voiture qui l'intéresse. Ensuite, la description et les valeurs financières apparaissent :

Description de la Ferrari 348

Type de Ferrari	Carrosserie	Type de motorisation	Puissance du moteur (en ch)	Date du début de fabrication du véhicule	Date de fin de fabrication du véhicule
TB	Coupé	V8	300	01/01/1989	-
TS	Targa	V8	-	-	-

Mis à jour le 02/12/08 à 16:56

Les valeurs financières de la Ferrari 348

Année de mise en circulation=1991

Type	Année de la vente aux enchères	Prix neuf du véhicule	Prix proposé lors de l'enchère	Prix obtenu lors de l'enchère	Evolution du prix depuis sa commercialisation (en %)	Délai de vente (en semestres)
		Moyenne	Moyenne	Moyenne	Moyenne	Moyenne
TB	2002	-	-	70222.85	-	22.00
TS	2002	-	-	38506.04	-	22.00

Figure 5 – Description et valeurs financières d'une voiture (ici Ferrari 348)

A partir de cette page de description et de valeurs financières, l'utilisateur peut retourner à la page précédente, tout en cliquant sur les différents logos de la marque, qui apparaissent sur cette interface.

4. CONCLUSION

4. CONCLUSION

Au cours de la réalisation de ce projet SAS, nous avons appliqué les différentes contraintes qui nous ont été impliquées. Ces différentes contraintes étaient de ne pas utiliser la procédure PRINT du module SAS/BASE, mais cela n'a pas été insurmontable. Car, nous avons l'opportunité de découvrir d'autres procédures qui pouvaient produire le même résultat. C'est trois procédures sont : la procédure TABULATE, la procédure REPORT et l'ODSOUT avec une étape DATA, du module SAS/BASE. L'utilisation de ces différentes procédures était aussi une contrainte d'obligation pour la réalisation de ce projet SAS. L'application de ces différentes procédures a été fastidieuse, car ces différentes procédures ont plusieurs avantages et inconvénients, qui ont été détaillés dans le chapitre 2.

Deux autres contraintes étaient fixées. Dans un premier temps, nous devons calculer la progression des valeurs des différentes voitures, ainsi que leur durée de conservation, par leur ancien propriétaire. Mais, cela ne s'est pas avéré trop difficile grâce à la procédure SQL ou avec l'aide d'une étape DATA, du module SAS/Base. Pour finir, sur les contraintes, nous devons conserver que les véhicules ayant une enchère et une description, ceci s'est réalisé avec l'aide de plusieurs requêtes SQL dans une procédure SQL, du module de SAS/Base.

Nous avons utilisé le module SAS/Macro, qui nous a permis de réaliser plusieurs rapports, comprenant les descriptions et les valeurs financières d'une voiture, selon les différentes voitures de collection présentes dans la base de données. Tous ces rapports ont été édités en pages web, avec l'objet ODS HTML, qui permettra de visualiser ces différents rapports par le biais de différents navigateurs web (Microsoft Internet Explorer, Mozilla Firefox, ...). Nous avons réalisé une feuille de style, pour la mise en forme de ces différentes pages web.

Pour conclure, l'application est enfin livrable. Mais, quelques modifications peuvent être encore apportées à ce projet. Il ne reste plus qu'à souhaité bon courage aux commerciaux de voitures de sport ; qui peuvent dès aujourd'hui, rassurer leur clientèle dans un investissement pérenne.

5. ANNEXES

5. ANNEXES

5.1 – Annexe 1 : Dictionnaire des données

Code	Libellé	Nature	Stabilité	Domaine	Commentaire
Marque	Marque de la voiture	Saisi	Stable	Texte	
Modèle	Modèle de la voiture	Saisi	Stable	Texte	
Type	Type de la voiture	Saisi	Stable	Texte	
Annee	Année de mise en circulation	Saisi	Stable	Date	
Carrosserie	Châssis	Saisi	Stable	Texte	
Type moteur	Type moteur	Saisi	Stable	Texte	
Turbo	Présence d'un turbo	Saisi	Stable	Booléen	
Cv_din	Puissance du moteur	Saisi	Stable	Entier [4]	
Vitesse	Vitesse maximale	Saisi	Stable	Entier [3]	
Nb_places	Nombre de places	Saisi	Stable	Entier [1]	
Deb_construction	Date de début de fabrication	Saisi	Stable	Date	
Fin_construction	Date de fin de fabrication	Saisi	Stable	Date	
Nb_exemplaire	Nombre d'exemplaire	Incrémenté	Stable/Evolutif	Entier [4]	
Prix_courant	Prix à la commercialisation	Saisi	Stable	Entier [10]	
Devise	Devise des prix	Saisi	Stable	Texte	
Ville	Ville de l'enchère	Saisi	Stable	Texte	
Date_vente	Date de vente	Saisi	Stable	Date	
Num_serie	Numéro de série	Saisi	Stable	Texte	
Conduite	Type de conduite	Saisi	Stable	Texte	
Kilometrage	Kilométrage	Saisi	Stable/Evolutif	Entier [6]	
Prix_propose	Mise de départ à l'enchère	Saisi	Stable/Evolutif	Entier [10]	

Adjugé	Prix vendu à l'enchère	Saisi	Stable	Entier [10]	
Qualité	Etat de la voiture	Saisi	Stable	Texte	
Commentaire	Commentaire	Saisi	Stable	Texte	
Annee_ip	Année de l'indice de prix	Incrémenté	Stable	Date	
Semestre	Semestre de l'indice de prix	Saisi	Stable	Texte	
Indice_prix	Indice de prix	Saisi	Stable	Numérique [3, 1]	
Délai	Délai de conservation	Calculé	Evolutif	Entier [3]	Année – Année de la date de vente
Progression	Progression de la valeur de la voiture	Calculé	Evolutif	Numérique [3, 1]	(Adjugé – Prix_courant)/Prix_courant

5.2 – Annexe 2 : Code du programme SAS

```
/*Gestion des dossiers sources et des dossiers de sorties.*/
%let chemin = D:\LP CSD\Projet SAS; /*Chemin à modifier avant de
l'exécution du programme.*/

libname auto "&chemin\Auto\";
filename entree "&chemin\Donnees Etudiants\";
filename sortie "&chemin\Sortie\";
filename image "&chemin\images\";
options fmtsearch = (auto work library) fmterr;

/*Le cours des devises à la date du 06 octobre 2008 à 10H45.*/
%let dm = 1.95583;
%let dollar = 0.7355;
%let livrest = 1.2928;
%let chf = 0.6452;
%let franc = 6.55957;

/*Recodage les modalités des variables Qualité et Conduite de la table
auto.encheres.*/
proc format;
    value $qualite
        "OR+", "OR-", "OR" = "Origine"
        "REST+", "REST-", "REST" = "Restaurée"
        "EX+", "EX-", "EX" = "Exceptionnelle"
        "RAR" = "Reste à restaurer"
    ;
    value $cond
        "RHD" = "Droite"
        "LHD" = "Gauche"
    ;
run;

/*Création de la table auto.encheres*/
data auto.encheres; /*Création de la table auto.encheres.*/
/*Importation du fichier Encheres.txt à partir de la deuxième ligne, ainsi
que les valeurs manquantes, le délimiteur est la virgule ",".*/
    infile entree(Encheres.txt) firstobs=2 missover dsd dlm=",";
/*Insertion de variables et définition de leurs tailles dans la table
auto.encheres.*/
    input ville :$30.
           date_v :$8.
           marque :$20.
           modele :$20.
           type :$50.
           annee :4.
           num_serie :$15.
           conduite :$10.
           kilometrage :6.
           prix_propose :6.
           adjuge :6.
           qualite :$15.
           commentaire :$100.
           devise :$10.
    ;
/*Redéfinition de la date de vente aux enchères, pour une meilleur
compréhension et pour la définir au format DATE.*/
    if length(date_v) = 8 then do;
```

```

        if substr(date_v, 5, 4) = '0000' then
            date_vente = input((substr(date_v, 1, 4) || '0630'),
YYMMDD8.);
        else
            date_vente = input(date_v, ANYDTDTE8.);
    end;
    else if length(date_v) = 6 then do;
        date_vente = input((substr(date_v, 1, 6) || '15'), YYMMDD8.);
    end;
    attrib date_vente format= DDMMYY10.;
/*Attribution de la modalit  Euros, pour les v hicules ayant un champ vide
dans la variable devise.*/
    if devise = "" then
        devise = 'Euros';
/*Conversion des prix propos s et des prix vendus en Euros.*/
    if devise = "DM" then
        propose = prix_propose / &dm;
    else if devise = "Livre ster" then
        propose = prix_propose * &livrester;
    else if devise = "Dollar US" then
        propose = prix_propose * &dollar;
    else if devise = "Euros" then
        propose = prix_propose;
    else if devise = "CHF" then
        propose = prix_propose * &chf;
    else if devise = "FF" then
        propose = prix_propose / &franc;
    if devise = "DM" then
        vendu = adjuge / &dm;
    else if devise = "Livre ster" then
        vendu = adjuge * &livrester;
    else if devise = "Dollar US" then
        vendu = adjuge * &dollar;
    else if devise = "Euros" then
        vendu = adjuge;
    else if devise = "CHF" then
        vendu = adjuge * &chf;
    else if devise = "FF" then
        vendu = adjuge / &franc;
    dev = "Euros";
/*Extraction de l'ann e de vente aux ench res.*/
    annee_vente = YEAR(date_vente);
/*Application de la proc dure FORMAT pr c demment.*/
    format conduite $cond.;
    format qualite $qualite.;
/*D finition de la taille des variables propose et vendu.*/
    format propose 15.0;
    format vendu 15.0;
/*Attribution de la modalit  Autre, pour les voitures qui ne poss dent pas
de type.*/
    if type = "" then type = "Autre";
/*Conserver uniquement les variables ci-dessous dans la table
auto.encheres.*/
    keep ville marque modele type annee num_serie conduite kilometrage
qualite commentaire date_vente annee_vente propose vendu dev;
run;

/*Cr ation de la table auto.description*/
data auto.description;
/*Importation du fichier auto.csv   partir de la deuxi me ligne, ainsi que
les valeurs manquantes,

```

```

le délimiteurs est le point-virgule ";"*/
infile entree(description auto.csv) firstobs=2 missover dsd dlm=";";
/*Insertion de variables et définition de leurs tailles dans la table
auto.description*/
input marque :$30.
      modele :$30.
      type :$20.
      carrosserie :$20.
      deb_construction :$6.
      fin_construction :$6.
      nb_exemplaire :5.
      type_moteur :$4.
      turbo :$3.
      cv_din :3.
      vitesse :3.
      nb_place :1.
;
/*Redéfinition de la date du début de fabrication de la voiture, pour une
meilleur compréhension et pour la définir au format DATE.*/
if length(deb_construction) = 6 then
      date_debut = input((substr(deb_construction, 1, 6) ||
'01'),YYMMDD8.);
else if length(deb_construction) = 4 then
      date_debut = input((substr(deb_construction, 1, 4) ||
'0101'),YYMMDD8.);
else date_debut=.;
/*Redéfinition de la date de fin de fabrication de la voiture, pour une
meilleur compréhension et pour la définir au format DATE.*/
if length(fin_construction) = 6 then
      date_fin = input((substr(fin_construction, 1, 6) ||
'15'),YYMMDD8.);
else if length(fin_construction) = 4 then
      date_fin = input((substr(fin_construction, 1, 4) ||
'1215'),YYMMDD8.);
else date_fin=.;
attrib      date_debut format = DDMMYY10.
            date_fin format = DDMMYY10.;
/*Attribuer le dernier jour du mois à la date_fin, pour déterminer la date
de fin de fabrication du véhicule.*/
      date_fin=intnx('MONTH',date_fin,0,'E');
drop deb_construction fin_construction;
/*Attribution de la modalité Autre, pour les voitures qui ne possèdent pas
de type.*/
if type = "" then type = "Autre";
run;

/*Création de la table auto.price*/
data auto.price;
/*Importation du fichier Vehicule.txt à partir de la deuxième ligne, ainsi
que les valeurs manquantes, le délimiteurs est la virgule ","*/
infile entree(Prix Vehicule.txt) firstobs=2 missover dsd dlm=",";
/*Insertion de variables et définition de leurs tailles dans la table
auto.price*/
input marque :$20.
      modele :$20.
      type :$20.
      annee :4.
      prix_courant :10.
      devise :$1.
;
/*Conversion du prix_courant en Euros.*/

```



```

        prix_courant = prix_courant / &franc;
        devise = '€';
/*Détermination de la taille de la variable prix_courant.*/
        format prix_courant 15.0;
/*Attribution de la modalité Autre, pour les voitures qui ne possèdent pas
de type.*/
        if type = "" then type = "Autre";
run;

/*Création de la table auto.ipconso*/
data auto.ipconso;
/*Importation du fichier IPConsommation base 100 1970.csv à partir de la
deuxième ligne, ainsi que les valeurs manquantes,
le délimiteurs est le point-virgule ";".*/
        infile entree(IPConsommation base 100 1970.csv) firstobs=2 missover
dsd dlm=";";
/*Insertion de variables et définition de leurs tailles dans la table
auto.ipconso.*/
        input annee :4.
                semestre :$2.
                indice_prix :numx10.
        ;
run;

/*Création de la table auto.ipconsommation, contenant les variables annee
et ip.
La variable ip est une moyenne de l'indice des prix en fonction de
l'année.*/
proc sql noprint;
        CREATE TABLE auto.ipconsommation AS SELECT DISTINCT annee,
AVG(indice_prix) AS ip
        FROM auto.ipconso GROUP BY annee;
/*Suppression de la table auto.ipconso*/
        DROP TABLE auto.ipconso;
quit;

/*Sélectionner l'indice prix le plus récent et le sauvegarder dans une
liste listip.*/
proc sql noprint;
        SELECT ip into: listip separated by "£"
        FROM auto.ipconsommation
        WHERE annee = (SELECT MAX(annee) FROM auto.ipconsommation);
quit;

/*Programmer une macro iprecent qui permet de stocker l'indice prix le plus
récent dans une macro-variable ip_act,
en fonction de la liste listip.*/
%macro iprecent;
        %do i = 1 %to 1;
        %global ip_act;
        %let ip_act = %scan(&listip, &i, "£");
        %end;
%mend;

/*Exécution de la macro iprecent.*/
%iprecent;

/*Afficher l'indice prix le plus récent dans le journal(log), pour
connaître sa valeur.*/
%put &ip_act;

```

```

/*Création de la table auto.enchere, contenant toutes les variables de la
table auto.encheres, sauf les variables propose et vendu,
qui ont été remplacées par prix_propose et prix_vendu, pour pouvoir
appliquer les indices prix en fonction de l'année sur une base 100
à partir de 1970.*/
proc sql noprint;
    CREATE TABLE auto.enchere AS SELECT ville, marque, modele, type,
encheres.annee, num_serie, conduite, kilometrage, qualite,
    commentaire, date_vente,
    (((proposé * 100) / ip) * &ip_act) / 100 AS prix_propose,
    (((vendu * 100) / ip) * &ip_act) / 100 AS prix_vendu, dev
    FROM auto.encheres, auto.ipconsommation
    WHERE annee_vente = ipconsommation.annee;
/*Suppression de la table auto.ipencheres.*/
DROP TABLE auto.encheres;
quit;

/*Création de la table auto.prix, contenant toutes les variables de la
table auto.encheres, sauf la variable prix_courant,
qui a été remplacée par prix_achat, pour pouvoir appliquer les indices prix
en fonction de l'année sur une base 100
à partir de 1970.*/
proc sql noprint;
    CREATE TABLE auto.prix AS SELECT marque, modele, type, price.annee,
    (((prix_courant * 100) / ip) * &ip_act) / 100 AS prix_achat,
    devise
    FROM auto.price, auto.ipconsommation
    WHERE price.annee = ipconsommation.annee;
/*Suppression de la table auto.price.*/
DROP TABLE auto.price;
quit;

/*Modification de la table auto.enchere, les caractères spéciaux ont été
remplacés par des espaces et le caractère é par un e,
et de supprimer tous les espaces dans les variables marque, modèle et type,
pour pouvoir créer une sorte de clé primaire nommé cle.*/
data auto.enchere;
    set auto.enchere;
    mark = COMPRESS(marque, "-");
    mark = COMPRESS(mark);
    model = COMPRESS(modele, "-");
    model = TRANWRD(model, "é", "e");
    model = COMPRESS(model);
    typ = TRANWRD(type, "-", " ");
    typ = TRANWRD(typ, "é", "e");
    typ = TRANWRD(typ, "/", " ");
    typ = TRANWRD(typ, "+", " ");
    typ = TRANWRD(typ, ".", " ");
    typ = TRANWRD(typ, ",", " ");
    typ = COMPRESS(typ);
    cle = TRIM(mark)||TRIM(model)||TRIM(typ);
run;

/*Modification de la table auto.description, les caractères spéciaux ont
été remplacés des espaces et le caractère é par un e,
et de supprimer tous les espaces dans les variables marque, modèle et type,
pour pouvoir créer une sorte de clé primaire nommé cle.*/
data auto.description;
    set auto.description;
    mark = COMPRESS(marque, "-");
    mark = COMPRESS(mark);

```

```

        model = COMPRESS(model, "-");
        model = TRANWRD(model, "é", "e");
        model = COMPRESS(model);
        typ = TRANWRD(type, "-", " ");
        typ = TRANWRD(typ, "é", "e");
        typ = TRANWRD(typ, "/", " ");
        typ = TRANWRD(typ, "+", " ");
        typ = TRANWRD(typ, ".", " ");
        typ = TRANWRD(typ, ",", " ");
        typ = COMPRESS(typ);
        cle = TRIM(mark)||TRIM(model)||TRIM(typ);
run;

/*Modification de la table auto.prix, les caractères spéciaux ont été
remplacés des espaces et le caractère é par un e,
et de supprimer tous les espaces dans les variables marque, modele et type,
pour pouvoir créer une sorte de clé primaire nommé cle.*/
data auto.prix;
    set auto.prix;
    mark = COMPRESS(marque, "-");
    mark = COMPRESS(mark);
    model = COMPRESS(model, "-");
    model = TRANWRD(model, "é", "e");
    model = COMPRESS(model);
    typ = TRANWRD(type, "-", " ");
    typ = TRANWRD(typ, "é", "e");
    typ = TRANWRD(typ, "/", " ");
    typ = TRANWRD(typ, "+", " ");
    typ = TRANWRD(typ, ".", " ");
    typ = TRANWRD(typ, ",", " ");
    typ = COMPRESS(typ);
    cle = TRIM(mark)||TRIM(model)||TRIM(typ);
run;

/*Trier la table auto.enchere par ordre alphabétique à partir de la
variable cle.*/
proc sort data=auto.enchere;
    by cle;
run;

/*Trier la table auto.description par ordre alphabétique à partir de la
variable cle.*/
proc sort data = auto.description;
    by cle;
run;

/*Trier la table auto.prix par ordre alphabétique à partir de la variable
cle.*/
proc sort data = auto.prix;
    by cle;
run;

/*Création de la table auto.cars qui est une jointure des tables
auto.description et auto.prix,
par la variable cle qui est une clé primaire. Conservation des variables
marque, modele, type, annee, carrosserie, type_moteur,
cv_din, date_debut, date_fin, prix_achat, devise, mark, model et cle.*/
data auto.cars;
    merge auto.description auto.prix;
    by cle;

```

```

    keep marque modele type annee carrosserie type_moteur cv_din
date_debut date_fin prix_achat devise mark model cle;
run;

/*Création de la table auto.voitures qui est une jointure des tables
auto.cars et auto.enchere,
par la variable cle qui est une clé primaire. Conservation des variables
marque, modele, type, annee, carrosserie, type_moteur,
cv_din, date_debut, date_fin, prix_achat, date_vente, prix_propose,
prix_vendu, dev, mark, model et cle.*/
data auto.voitures;
    merge auto.cars auto.enchere;
    by cle;
    keep marque modele type annee carrosserie type_moteur cv_din
date_debut date_fin prix_achat date_vente prix_propose
    prix_vendu dev mark model cle;
run;

/*Modification de la table auto.voitures, en ajoutant la variable
progression qui calcule la progression du prix vente du véhicule
à l'enchère par rapport au prix neuf de la voiture. La variable delai
correspond à la différence de temps entre l'année du véhicule
et l'année de vente aux enchères, calculée en semestre.*/
data auto.voitures;
    set auto.voitures;
    progression = (((prix_vendu - prix_achat) / prix_achat) * 100);
    delai = ((YEAR(date_vente) - annee) * 2);
    annee_vente = YEAR(date_vente);
/*Attribution de la modalité 0, pour les voitures ne possédant pas d'année
de vente aux enchères.*/
    if annee_vente = "" then annee_vente = 0;
/*Attribution de la modalité 0, pour les voitures ne possédant pas d'année
de mise en circulation.*/
    if annee = "" then annee = 0;
/*Attribution de la modalité Inconnu, pour les voitures ne possédant pas de
carrosserie.*/
    if carrosserie = "" then carrosserie = "Inconnu";
/*Attribution de la modalité Inconnu, pour les voitures ne possédant pas de
moteur.*/
    if type_moteur = "" then type_moteur = "Inconnu";
/*Attribution de la modalité 0, pour les voitures n'ayant pas de puissance
au niveau moteur.*/
    if cv_din = "" then cv_din = 0;
/*Attribution de la modalité 0, pour les voitures ne possédant pas de date
de début de fabrication.*/
    if date_debut = "" then date_debut = 0;
/*Attribution de la modalité 0, pour les voitures ne possédant pas de date
de fin de fabrication.*/
    if date_fin = "" then date_fin = 0;
/*Détermination de la taille des variables prix_propose, prix_vendu,
prix_achat, progression et delai.*/
    format prix_propose 15.;
    format prix_vendu 15.;
    format prix_achat 15.;
    format progression 5.;
    format delai 3.;
run;

proc sql noprint;
/*Suppression des voitures n'ayant pas été vendu aux enchères, et n'ayant
pas de description.*/

```

```

DELETE FROM auto.voitures WHERE annee_vente = 0;
DELETE FROM auto.voitures WHERE annee = 0;
DELETE FROM auto.voitures WHERE carrosserie = "Inconnu"
AND type_moteur = "Inconnu"
AND cv_din = 0
AND date_debut = 0
AND date_fin = 0;
quit;

data auto.voitures;
    set auto.voitures;
/*Recodage des variables, de façon que ce soit compréhensible.*/
    if carrosserie = "Inconnu" then carrosserie = "";
    if type_moteur = "Inco" then type_moteur = "";
    if cv_din = 0 then cv_din = "";
    if date_debut = 0 then date_debut = "";
    if date_fin = 0 then date_fin = "";
run;

proc sql noprint;
/*Création de la table auto.marque à partir de la table auto.voitures,
contenant seulement les variables marque et mark,
groupé par marque.*/
    CREATE TABLE auto.marque AS SELECT DISTINCT marque, mark
    FROM auto.voitures
    ORDER BY marque;

/*Création de la table auto.modele à partir de la table auto.voitures,
contenant seulement les variables marque, mark, modele et model,
groupé par marque.*/
    CREATE TABLE auto.modele AS SELECT DISTINCT marque, mark, modele,
model
    FROM auto.voitures
    ORDER BY marque;

/*Suppression de la table auto.cars.*/
    DROP TABLE auto.cars;
quit;

/*Création de la table auto.formatmarque à partir de la table auto.marque,
qui permettra de renommer les modalités de la variable marque.*/
data auto.formatmarque;
    set auto.marque;
/*Attribution et nomination de variables dans la table auto.formatmarque.*/
    attrib START
        LABEL = "START";
    attrib LABEL
        LABEL = "LABEL";
    attrib FMTNAME
        LABEL = "FMTNAME";
    attrib TYPE
        LABEL = "TYPE";
/*Nom de la procédure FORMAT qui sera appliqué à la variable marque.*/
    FMTNAME = "markvoiture";
/*Insertion des différentes modalités de la variable marque dans la
variable START.*/
    START = marque;
/*Détermination du type alphanumérique pour le format de la variable
marque.*/
    TYPE = "C";

```

```

/*Codage des modalités de la variable marque en lien hypertexte, pour les
sorties en html.*/
    LABEL = '<A HREF="' || COMPRESS(mark) || '.html">' || TRIM(START) || "</a>";
run;

/*Création de la table auto.formatmodele à partir de la table auto.modele,
qui permettra de renommer les modalités de la variable modele.*/
data auto.formatmodele;
    set auto.modele;
/*Attribution et nomination de variables dans la table auto.formatmodele.*/
    attrib START
        LABEL = "START";
    attrib LABEL
        LABEL = "LABEL";
    attrib FMTNAME
        LABEL = "FMTNAME";
    attrib TYPE
        LABEL = "TYPE";
/*Insertion des différentes modalités de la variable modele dans la
variable START.*/
    START = modele;
/*Nom de la procédure FORMAT qui sera appliqué à la variable modele.*/
    FMTNAME = "modelevoiture";
/*Détermination du type alphanumérique pour le format de la variable
modele.*/
    TYPE = "C";
/*Codage des modalités de la variable marque en lien hypertexte, pour les
sorties en html.*/
    LABEL = '<A HREF="' || COMPRESS(mark) || COMPRESS(model) || '.html">'
|| TRIM(START) || "</a>";
run;

/*Création de la table auto.formatvoiture par la jointure des tables
auto.formatmarque et auto.formatmodele.*/
data auto.formatvoiture;
    set auto.formatmarque auto.formatmodele;
run;

/*Application de la définition de nos propres formats pour la lecture et
pour l'écriture des valeurs des variables marque et modele.*/
proc format CNTLIN = auto.formatvoiture LIBRARY = auto;
run;

/*Modification de la table auto.marque, pour stocker le chemin des logos
des différentes marques et pour l'affectation de lien hypertexte.*/
data auto.marque;
    set auto.marque;
    lien_logo = ("<A href =" || TRIM(mark) || ".html"><img border='0' src =
'../images/" || TRIM(marque) || ".png"></A>");
run;

/*Trier la table auto.marque par ordre alphabétique de la variable
marque.*/
proc sort data = auto.marque;
    by marque;
run;

/*Création de la page web Voiture.html en y appliquant notre propre feuille
de style.*/
ods html path = sortie file="Voiture.html" (title = "Marques des voitures")
stylesheet = (url = 'style.css');

```

```

/*Titre de la page web.*/
title "Liste des marques de voitures";
data auto.vehicule;
set auto.marque end=fin_fic;
by marque;
/*On déclare l'objet ODSOUT sur la 1ère observation avant le début de la
boucle qu'est une étape DATA.*/
if _n_=1 then do;
/*L'objet ODSOUT doit avoir un nom. Pour le projet, c'est mis_form.*/
declare odsout mis_form();
/*On construit un tableau.*/
mis_form.table_start();

end;
if first.marque then do;
/*On ajoute une ligne dans le tableau.*/
mis_form.row_start();
mis_form.row_end();
/*On rajoute une ligne dans le tableau. Cette ligne est de type en-tête
(HEADING).*/
mis_form.row_start(type:"HEADING");
/*Dans cette ligne, on remplit les cellules contenant les différentes
modalités de la variable lien_logo.*/
mis_form.format_cell(text:lien_logo,

OVERRIDES:"JUST=CENTER", OVERRIDES:"VJUST=MIDDLE");
/*Dans cette ligne, on remplit les cellules contenant les différentes
modalités de la variable marque,
en y appliquant le format créé précédemment.*/
mis_form.format_cell(text:put(marque,
$markvoiture.),

OVERRIDES:"JUST=CENTER", OVERRIDES:"VJUST=MIDDLE");
mis_form.row_end();
end;
/*Fin du tableau de l'ODSOUT.*/
if fin_fic then mis_form.table_end();
/*Information de mise à jour dans le pied page de la page web.*/
footnote "Mis à jour le %sysfunc(today()), ddmm yy) à
%sysfunc(time()), hhmm)";
run;
/*Fermeture de la page web Voiture.html.*/
ods html close;

/*Sélectionner les marques de voitures de la variable marque de la table
auto.voitures et les stocker dans la liste listmarque.*/
proc sql noprint;
SELECT DISTINCT marque into: listmarque separated by "&"
FROM auto.voitures;
quit;

/*Sauvegarder le nombre de marques de voitures dans la macro-variable
maxmarque.*/
%let maxmarque = &sqllobs;

/*Sélectionner les marques de voitures de la variable mark de la table
auto.voitures et les stocker dans la liste listmark.*/
proc sql noprint;
SELECT DISTINCT mark into: listmark separated by "&"
FROM auto.voitures;
quit;

```

```

/*Création de la macro voiture.*/
%macro voiture;
/*Boucle j partant de 1 jusqu'au nombre de marques différentes.*/
%do j = 1 %to &maxmarque;

/*Stockage des marques listées dans la liste listmarque, dans la macro-
variable marque.*/
%let marque = %scan(&listmarque, &j, "£");
/*Stockage des marque listées dans la liste listmark, dans la macro-
variable mark.*/
%let mark = %scan(&listmark, &j, "£");

/*Création des tables auto.&mark à partir de la table auto.voitures,
sélectionnant des modèles de voitures à partir de la variable marque, en
fonction de la marque desvoitures.*/
proc sql;
CREATE TABLE auto.&mark AS SELECT DISTINCT modele
FROM auto.voitures
WHERE marque = "&marque";
quit;

/*Création des pages web &mark.html en y appliquant notre propre feuille de
style.*/
ods html path = sortie file = "&mark..html" (title = "&marque")
stylesheet = (url = 'style.css');
/*Utilisation de le procédure REPORT pour les tables auto.&mark, avec
interdiction d'ouvrir une fenêtre indépendante de windows,
soulignez tous les en-têtes de colonnes et inscription d'une interligne
sous toutes les en-têtes de colonnes.*/
proc report data = auto.&mark nowindows headline
headskip;
/*Titres des pages web &mark.html.*/
title "Liste des modèles des voitures &marque";
title2;
title3 "<img src = '../images/&marque..png'>";
/*Création d'un tableau avec une seule colonne comprenant les valeurs de la
variable modele des tables auto.&mark.*/
columns modele;
/*Renommer l'en-tête de la colonne modele dans le tableau.*/
define modele / display "Modèle";
/*Recodage des modalités de la variable modele comprenant les liens
hypertextes, avec l'aide la procédure Format CNTLIN réalisée
précédemment.*/
format modele $modelevoitu.;
/*Information de mise à jour et de lien de redirection vers le fichier
racine Voiture.html dans le pied page de la page web.*/
footnote "<A HREF = 'Voiture.html'>Liste des marques
de voitures</A>";
footnote2 "Mis à jour le %sysfunc(today(), ddmmyy)
à %sysfunc(time(), hhmm)";
run;
/*Fermeture des pages web &mark.html.*/
ods html close;

/*Modification des tables auto.&mark à partir de la table auto.voitures,
sélection des modèles de voitures à partir des variables modele et model,
en fonction de la marque des voitures.*/
proc sql noprint;
CREATE TABLE auto.&mark AS SELECT modele, model
FROM auto.voitures
WHERE marque = "&marque";

```



```

/*Sélectionner les modèles de voitures de la variable model des tables
auto.&mark et de les stocker dans la liste listmodel,
en fonction de la marque des voitures.*/
        SELECT DISTINCT model into: listmodel separated by "£"
        FROM auto.&mark;
/*Sélectionner les modèles de voitures de la variable modele des tables
auto.&mark et de les stocker dans la liste listmodele,
en fonction de la marque des voitures.*/
        SELECT DISTINCT modele into: listmodele separated by "£"
        FROM auto.&mark;
quit;

/*Sauvegarder le nombre de modèles de voitures dans la macro-variable
maxmodele, en fonction de la marque de la voiture.*/
        %let maxmodele = &sqllobs;

/*Modification des tables auto.&mark à partir de la table auto.voitures,
comprenant uniquement les variables qui nous intéressent,
en fonction de la marque des voitures.*/
        proc sql noprint;
                CREATE TABLE auto.&mark AS SELECT modele, type, annee,
carrosserie, type_moteur, cv_din, date_debut, date_fin, prix_achat,
                date_vente, annee_vente, prix_propose, prix_vendu, dev,
progression, delai
                FROM auto.voitures
                WHERE marque = "&marque";
        quit;

/*Boucle k partant de 1 jusqu'au nombre de modèles différents de la marque
des voitures.*/
        %do k = 1 %to &maxmodele;

/*Stockage des modèles listées dans la liste listmodele, dans la macro-
variable modele, en fonction de la marque des voitures.*/
                %let modele = %scan(&listmodele, &k, "£");
/*Stockage des modèles listées dans la liste listmodel, dans la macro-
variable model, en fonction de la marque des voitures.*/
                %let model = %scan(&listmodel, &k, "£");

/*Création des tables auto.&mark&model à partir des tables auto.&mark,
sélectionnant uniquement les variables qui nous intéressent pour faire la
description d'une voiture,
en fonction de la marque et du modèle des voitures. Regrouper par type.*/
        proc sql noprint;
                CREATE TABLE auto.&mark&model AS SELECT DISTINCT
type, carrosserie, type_moteur, cv_din,
                date_debut, date_fin
                FROM auto.&mark
                WHERE modele = "&modele"
                ORDER BY type;

/*Création des tables auto.&mark&model&model à partir des tables
auto.&mark,
sélectionnant uniquement les variables qui nous intéressent pour décrire
les valeurs financières d'une voiture,
en fonction de la marque et du modèle des voitures. Regrouper par type.*/
        CREATE TABLE auto.&mark&model&model AS SELECT
DISTINCT type, annee, prix_achat, annee_vente, prix_propose, prix_vendu,
                progression, delai
                FROM auto.&mark
                WHERE modele = "&modele"

```

```

ORDER BY annee;
quit;

/*Création des pages web &mark&model.html en y appliquant notre propre
feuille de style.*/
ods html path = sortie file = "&mark&model..html" (title
= "&marque &modele") stylesheet = (url = 'style.css');
/*Titres des pages web &mark&model.html.*/
title " Description de la &marque &modele";
title2;
title3 "<A href = '&mark..html'><img border='0' src
= '../images/&marque..png'></A>";

/*Utilisation de le procédure REPORT pour les tables auto.&mark&model, avec
interdiction d'ouvrir une fenêtre indépendante de windows,
soulignez tous les en-têtes de colonnes et inscription d'une interligne
sous toutes les en-têtes de colonnes.*/
proc report data = auto.&mark&model nowindows
headline headskip;
/*Création d'un tableau a six colonnes comprenant les valeurs des variable
type, carrosserie, type_moteur, cv_din, date_debut et date_fin
des tables auto.&mark&model.*/
columns type carrosserie type_moteur cv_din
date_debut date_fin;
/*Renommer les en-têtes des colonnes type, carrosserie, type_moteur,cv_din,
date_debut et date_fin dans le tableau.*/
define type / display "Type de &marque
&modele";
define carrosserie / display "Carrosserie";
define type_moteur / display "Type de
motorisation";
define cv_din / display "Puissance du moteur
(en ch)";
define date_debut / display "Date du début de
fabrication du véhicule";
define date_fin / display "Date de fin de
fabrication du véhicule";
footnote "Mis à jour le %sysfunc(today(),
ddmmyy) à %sysfunc(time(),hhmm)";
run;

/*Utilisation de le procédure TABULATE pour les tables
auto.&mark&model&model.*/
proc tabulate data = auto.&mark&model&model;
/*Autre titre des pages web &mark&model.html.*/
title " Les valeurs financières de la &marque
&modele";
/*Renommer les en-têtes des colonnes type, annee, prix_achat, annee_vente,
prix_propose, prix_vendu, progression et delai
du tableau statistique.*/
label type = "Type"
annee = "Année de mise en
circulation"
prix_achat = "Prix neuf du
véhicule"
annee_vente = "Année de la vente
aux enchères"
prix_propose = "Prix propose lors
de l'enchère"
prix_vendu = "Prix vendu lors de
l'enchère"


```

```

                                progression = "Evolution du prix
depuis sa commercialisation (en %)"                                delai = "Délai de vente (en
semesestres)";
/*Liste des variables d'analyse sur lesquelles sont effectuées les calculs
statistiques.*/
                                var prix_achat prix_propose prix_vendu
progression delai;
/*Liste des variables de classe définissant les pages, les lignes et les
colonnes.*/
                                class type annee annee_vente;
/*Définir la construction du tableau statistique, sachant que le tableau
peut avoir au maximum trois dimensions.
Avec insertion du logo de la marque dans la boîte vide au-dessus des titres
de lignes du tableau statistique,
avec lien de redirection vers la liste des modèles de la marque des
voitures, à partir du logo de la marque.*/
                                table type * annee_vente, prix_achat * mean
prix_propose * mean prix_vendu * mean progression * mean delai * mean /
                                BOX = "<A href ='&mark..html'><img border='0'
src = '../images/&marque..png'></A>";
                                by annee;
/*Renommer les en-têtes aux mots clés définissant les statistiques.*/
                                keylabel MEAN = "Moyenne";
/*Information de mise à jour.*/
                                footnote "Mis à jour le %sysfunc(today(),
ddmmyy) à %sysfunc(time(),hhmm)";
                                run;
/*Fermeture des pages web &mark&model.html.*/
                                ods html close;
/*Fin de la boucle k.*/
                                %end;
/*Fin de la boucle j.*/
                                %end;
/*Fermeture de la macro voiture.*/
%mend;
/*Exécution de la macro voiture.*/
%voiture;

```

5.3 – Annexe 3 : Architecture de l'application web



5.4 – Annexe 4 : Références

Les livres

KONTCHOU-KOUOMEGNI H. & DECOURT O., *Maîtriser SAS Base, SAS Macro, SAS 9.2 et versions antérieures*. DUNOD, Paris, 2007, 256 p.

Les articles

EDUCASOFT FORMATION, *L'ODS avec SAS 9*, 48 p.

EDUCASOFT FORMATIONS, *Nouveautés SAS version 9*, 135 p.

EDUCASOFT FORMATIONS, *Initiation à SAS*, 161 p.

Les informations sur Internet

Moteur de recherche : Google > <http://www.google.fr>

Consultation du support SAS : Support SAS > <http://support.sas.com/>

Site web : Olivier DECOURT SARL > <http://www.od-datamining.com>

Blog sur SAS : La référence SAS > <http://thesasreference.wordpress.com>

Site web : UCL Institut de statistique > <http://www.stat.ucl.ac.be/SMCS/serveur/sas.html>